
! 
! Magnolia CSL model file created on 2019-04-18T10:21:02.815 
! PBPK Gas Uptake Model for 2,2-DCPe converted from original acslx model by EMK 
! Units of this model are liters (L), hours (h), milligrams (mg), BW in kilograms (kg) 
 
model DCP22 
 
initial  ! The INITIAL section contains statements evaluated 1x @ begin simulation 
 
!***********! Physiological Parameters**********************} 
CONSTANT     QPC = 13.2               ! Alveolar Ventilation Rate(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264 
CONSTANT     QCC = 13.2              ! Cardiac Output(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264 
CONSTANT     QLC = 0.242            ! Fract. Blood Flow to Liver, Table 25, p. 439, Brown et al., 1997 
CONSTANT     QFC = 0.082             ! Fract. Blood Flow to Fat, Table 25, p. 439, Brown et al., 1997 
CONSTANT     QSC = 0.257            ! Fract. Blood Flow to Poorly Perfused T., , Table 25, p. 439, Brown et al., 1997 (muscle + 
skin) 
             QRC = 1.0 - QLC - QFC - QSC ! Fract. Blood Flow to Richly Perfused T 
 
CONSTANT    BW = 0.222               ! Body Weight(kg), experiment-specific, mean & SD all rats 0.222+/-0.0184  
CONSTANT   VLC = 0.036               ! Volume Fraction Liver, Brown et al., 1997, Table 8, p. 418 
CONSTANT   VFC = 0.076             ! Volume Fraction Fat, Brown et al., 1997, equation, p. 422 using avg BW  
CONSTANT   VSC = 0.7               ! Volume Fraction Poorly Perfused T., muscle + skin, Table 5, p. 416, Brown et al., 1997 
           VRC = 1.0 - VLC - VFC - VSC      ! Volume Fraction Richly Perfused T. 
 
  !************! Chemical Specific Parameters for DCPs************************} 
  ! Partition Coeffs Estimated Based on method of Beliveau et al., (2003)} 
CONSTANT        PL =  1.81       ! Liver/Blood 
CONSTANT        PF  = 22.03      ! Fat/Blood 
CONSTANT        PS  = 0.84       ! Poorly/Blood (muscle is representative tissue) 
CONSTANT        PR  = 1.81       ! Richly/Blood (liver is representative tissue) 
CONSTANT        PB  = 3.75       ! Blood/Air 
CONSTANT        MW  = 112.99     ! Molecular Wt (g/mol) 

!VmaxC & Km starting values - final values are estimated from set of vapor uptake curves 
!CONSTANT      VmaxC = 4.46   ! mg/hr-kg 
!CONSTANT         Km = 3.77   ! mg/L 
CONSTANT      KFC = 1.18  ! L/hr-kg, use if assuming first order metabolism 
 
 ! Chamber/Rat/Exposure Parameters} 
CONSTANT       CONC = 500.            ! Init. Inhaled Conc (ppm), experiment-specific 
CONSTANT       RATS = 1               ! Number of rats in chamber, always 1 
CONSTANT       VCHC = 3.8              ! Closed chamber vol (l) 
CONSTANT       KLOS = 0.04              ! Closed chamber loss, hr-1, experiment-specific 
     NETVOL = VCHC - (RATS * BW)            ! Net Chamber Vol (l), i.e. volume of chamber not taken up by rats 
        AIO = (CONC * NETVOL * MW)/24450       ! Initial Conc of chemical in Chamber (mg/l) 
CONSTANT   TSTOP = 6.00                          ! Length of experiment (hrs) 
CONSTANT  points = 100.                          ! No. of points in plot 
     
!*******Scaled Parameters********} 
   
         QC = QCC * (BW**0.75)      ! Cardiac Output (l/hr) 
         QP = QPC * (BW**0.75)      ! Alveolar Ventilation Rate (l/hr) 
         QL = QLC * QC              ! Flow Liver Compartment (l/hr) 
         QF = QFC * QC           ! Flow Fat Compartment (l/hr) 
         QS = QSC * QC              ! Flow Slowly Perf. Tis. Cmpt. (l/hr) 
         QR = QRC * QC              ! Flow Richly Perf. Tis. Cmpt. (l/hr) 
         VL = VLC * BW              ! Volume Liver Compartment, Total 
         VF = VFC * BW              ! Volume Fat Compartment 
         VS = VSC * BW              ! Volume Slowly Perfused Tis. Cmpt. 
         VR = VRC * BW              ! Volume Richly Perfused Tis. Cmpt. 
 
!Comment in/out eq below if running w/ michaelis-menten vs. first order} 
  !VMAX = VMAXC * BW**0.75    ! VMAX scaled 
 KF = KFC * BW**0.75       !      
end ! initial 
 
dynamic   !  The DYNAMIC section statements evaluated at each output time point     

    derivative         
 
        RAI = (RATS * QP * (CX - CI)) - (KLOS*AI)       ! mg/hr 
         AI = INTEG(RAI, AIO)                           ! mg 
         CI = AI / NETVOL                      ! mg/l 
         CP = (CI* 24450/MW)                   ! chamber conc, ppm, data is in ppm 
    
! chamber loss 
      RLOSS = (KLOS * AI)                               ! mg/hr 
       LOSS = INTEG(RLOSS, 0.0) 
     
    CA = ((QC * CV) + (QP * CI)) / (QC + (QP / PB))  ! arterial (mg/L) 
  
 !exhaled breath 
        RAX = QP * CX                                   ! mg/hr 
         AX = INTEG(RAX, 0.0)                           ! mg 
         CX = CA/PB                          ! Conc. DCPe in exhaled air(mg/l) 
      CXPPM = ((0.7 * CX) + (0.3 * CI)) * (24.45/MW)   ! Conc. DCPE in ppm  exhaled breath  
         AXKG = AX/BW                        ! mg exhaled/kg body weight 
  
!slowly perfused tissue group 
    RAS = QS * (CA - CVS)                ! mg/hr 
     AS = INTEG(RAS,0.0)                 ! mg 
    CVS = AS / (VS * PS)        ! Conc partition to slow per. tis.(mgl) 
    CS  = AS / VS                ! Conc in volume slow per. tis.(mg/l) 
 
!rapidly perfused tissue group 
    RAR = QR * (CA - CVR)                ! mg/hr                
     AR = INTEG(RAR, 0.0)                ! mg 
    CVR = AR / (VR * PR)          ! Conc partition to rap per. tis.(mg/l) 
     CR = AR / VR                   ! Conc in volume rap per. tis.(mg/l) 
  
!fat  

    RAF = QF * (CA - CVF)               ! mg/hr 
     AF = INTEG(RAF, 0.0)               ! mg 
    CVF = AF / (VF * PF)       ! Conc partition to fat(mg/l) 
    CF  = AF / VF                ! Conc in fat volume(mg/l) 
  
!liver 
    RAL = QL * (CA - CVL) - RAM         ! mg/hr 
     AL = INTEG(RAL, 0.0)               ! mg 
    CVL = AL / (VL * PL)       ! Conc partition to liver(mg/l)  
     CL = AL / VL               ! Conc in liver volume(mg/l) 
  
!Note: RAM rate of metabolism can be switched between being expressed as  
!Michaelis-Menten terms (mg/hr) and Clearance terms (L/hr) by commenting 
!in or out between the two equations below.  Note that parameters need to 
!be changed within the inital section also (2 places) VMAX vs. KF & VMAXC vs KFC} 
    
!    RAM   = (VMAX * CVL) / (KM + CVL) 
    RAM = KF * CVL    
     AM = INTEG(RAM, 0.0) 
 
    CV = (QF*CVF + QL*CVL + QS*CVS + QR*CVR) / QC    ! venous blood conc (mg/l) 
 
    DOSE = AIO - AI                 ! amt dcp in rat (mg) 
  
    DMASS = (AF + AL + AS + AR + AM)*RATS 
     
!Criterion - mass balance (MB)  should be near zero, i.e. <= ~e10-8.} 
     
    MB = DOSE - DMASS - LOSS            ! mass balance (mg)       
     
    TERMT(T >= TSTOP)                   ! Condition for terminating simulation 
  
end ! derivative     

end ! dynamic 
 
end ! program 

! Parameter estimation script for 2,2-DCP PBPK model 
! Closed-chambed rat data, 2,2-dichloropropane 
 
load 'DCP22.csl' 
 
! Any or all but one individal line can be commented out to simulate a single expt. 
! Each line below calls data file and species parameters that change with each experiment 
data @file='dcp2_60a.csv' ds1  t='t' cp='cp' tstop=4.17 bw=0.217 klos=0.0305 conc=49 
data @file='dcp2_60b.csv' ds2  t='t' cp='cp' tstop=4.17 bw=0.222 klos=0.0332 conc=50 
data @file='dcp2_60c.csv' ds3  t='t' cp='cp' tstop=5.00 bw=0.214 klos=0.0208 conc=57 
data @file='dcp2_60d.csv' ds4  t='t' cp='cp' tstop=5.00 bw=0.209 klos=0.0312 conc=60 
data @file='dcp2_200a.csv' ds5  t='t' cp='cp' tstop=4.33 bw=0.213 klos=0.045 conc=190 
data @file='dcp2_200b.csv' ds6  t='t' cp='cp' tstop=4.33 bw=0.226 klos=0.025 conc=183 
data @file='dcp2_230a.csv' ds7  t='t' cp='cp' tstop=4.17 bw=0.211 klos=0.043 conc=221 
data @file='dcp2_500a.csv' ds8  t='t' cp='cp' tstop=6.0 bw=0.231 klos=0.032 conc=454 
data @file='dcp2_500b.csv' ds9  t='t' cp='cp' tstop=6.1 bw=0.267 klos=0.036 conc=496 
data @file='dcp2_600a.csv' ds10  t='t' cp='cp' tstop=5.67 bw=0.219 klos=0.059 conc=586 
data @file='dcp2_600b.csv' ds11  t='t' cp='cp' tstop=6.17 bw=0.210 klos=0.065 conc=617 
data @file='dcp2_1150a.csv' ds12 t='t' cp='cp' tstop=5.92 bw=0.189 klos=0.0468 conc=1105 
data @file='dcp2_1150b.csv' ds13 t='t' cp='cp' tstop=6.1 bw=0.266 klos=0.063 conc=1141 
data @file='dcp2_1150c.csv' ds14 t='t' cp='cp' tstop=6.33 bw=0.248 klos=0.0035 conc=1242 
data @file='dcp2_1300a.csv' ds15 t='t' cp='cp' tstop=5.92 bw=0.197 klos=0.0518 conc=1288 
 
! A procedure to plot all data sets 
! Used to plot the fits before and after parameter estimation 
procedure plotall 
 
    set tstop=4.17 bw=0.217 klos=0.0305 conc=49 
    start 
    plot cp 'ds1:cp' @title='2,2-DCP 60a' 
     
    set tstop=4.17 bw=0.222 klos=0.0332 conc=50 
    start 
    plot cp 'ds2:cp' @title='2,2-DCP 60b' 

   set tstop=5.00 bw=0.214 klos=0.0208 conc=57 
   start 
   plot cp 'ds3:cp' @title='2,2-DCP 60c' 
     
    set tstop=5.00 bw=0.209 klos=0.0312 conc=60 
    start 
    plot cp 'ds4:cp' @title='2,2-DCP 60d' 
     
    set tstop=4.33 bw=0.213 klos=0.045 conc=190 
    start 
    plot cp 'ds5:cp' @title='2,2-DCP 200a' 
     
    set tstop=4.33 bw=0.226 klos=0.025 conc=183 
    start 
    plot cp 'ds6:cp' @title='2,2-DCP 200b' 
     
    set tstop=4.17 bw=0.211 klos=0.043 conc=221 
    start 
    plot cp 'ds7:cp' @title='2,2-DCP 230a' 
     
    set tstop=6.0 bw=0.231 klos=0.032 conc=454 
    start 
    plot cp 'ds8:cp' @title='2,2-DCP 500a' 
     
    set tstop=6.1 bw=0.267 klos=0.036 conc=496 
    start 
    plot cp 'ds9:cp' @title='2,2-DCP 500b' 
      
    set tstop=5.67 bw=0.219 klos=0.059 conc=586 
    start 
    plot cp 'ds10:cp' @title='2,2-DCP 600a' 
 
    set tstop=6.17 bw=0.210 klos=0.065 conc=617 

    start 
    plot cp 'ds11:cp' @title='2,2-DCP 600b' 
     
    set tstop=5.92 bw=0.189 klos=0.0468 conc=1105 
    start 
    plot cp 'ds12:cp' @title='2,2-DCP 1150a' 
     
    set tstop=6.1 bw=0.266 klos=0.063 conc=1141 
    start 
    plot cp 'ds13:cp' @title='2,2-DCP 1150b' 
     
    set tstop=6.33 bw=0.248 klos=0.0035 conc=1242 
    start 
    plot cp 'ds14:cp' @title='2,2-DCP 1150c' 
     
    set tstop=5.92 bw=0.197 klos=0.0518 conc=1288 
    start 
    plot cp 'ds15:cp' @title='2,2-DCP 1300a' 
 
end 
 
 
prepare @clear T CP 
 
set kfc = 1.08 @min=0.1 @max=2.16 
 
plotall 
 
fit @method=ml @errormodel=mixed kfc 
 
plotall 
  



9 
 

! 
! Magnolia CMD script created on 2020-02-24T16:12:57.165 
! Sensitivity Analysis 22-DCP Global Morris Method 
 
load 'DCP22.csl' 
 
set     QPC = 13.2  @min=10 @max=16   ! Alveolar Ventilation Rate(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264 
set     QCC = 13.2  @min=10 @max=16     ! Cardiac Output(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264 
set     QLC = 0.242 @min=0.2 @max=0.28 !Fract. Blood Flow to Liver, Table 25, p. 439, Brown et al., 1997 
set     QFC = 0.082 @min=0.05 @max=0.11 ! Fract. Blood Flow to Fat, Table 25, p. 439, Brown et al., 1997 
set     QSC = 0.257 @min=0.23 @max=0.28 ! Fract. Blood Flow to Poorly Perfused T., , Table 25, p. 439, Brown et al., 1997 (muscle + skin) 
 
set   BW = 0.222    @min=0.19 @max=0.25   ! Body Weight(kg), experiment-specific, mean & SD all rats 0.222+/-0.0184  
set   VLC = 0.036   @min=0.028 @max=0.044 ! Volume Fraction Liver, Brown et al., 1997, Table 8, p. 418 
set   VFC = 0.076   @min=0.05 @max=0.09   ! Volume Fraction Fat, Brown et al., 1997, equation, p. 422 using avg BW  
set   VSC = 0.7     @min=0.65 @max=0.75   ! Volume Fraction Poorly Perfused T., muscle + skin, Table 5, p. 416, Brown et al., 1997 
 
  !************! Chemical Specific Parameters for DCPs************************} 
  ! Partition Coeffs Estimated Based on method of Beliveau et al., (2003)} 
set        PL =  1.81    @min=1.0 @max=2.6   ! Liver/Blood 
set        PF  = 22.03   @min=18 @max=26   ! Fat/Blood 
set        PS  = 0.84    @min=0.6 @max=1.2   ! Poorly/Blood (muscle is representative tissue) 
set        PR  = 1.81    @min=1.0 @max=2.6   ! Richly/Blood (liver is representative tissue) 
set        PB  = 3.75    @min=2.8 @max=4.7   ! Blood/Air 
set        MW  = 112.99     ! Molecular Wt (g/mol) 
 
!VmaxC & Km starting values - final values are estimated from set of vapor uptake curves 
set      KFC = 1.18  @min=0.1 @max=2.2   ! L/hr-kg, hepatic clearance 
 
! Chamber/Rat/Exposure Parameters} 
set       CONC = 50.       ! Init. Inhaled Conc (ppm), experiment-specific, use 50, 200, 500 and 1150 ppm 
set       RATS = 1      ! Number of rats in chamber, always 1 
set       VCHC = 3.8     ! Closed chamber vol (l) 
set       KLOS = 0.04 @min=0.02 @max=0.06 ! Closed chamber loss, hr-1, experiment-specific 

set      TSTOP = 6.00       ! Length of exposure 
 
prepare t, cp, mb 
 
output @clear t cp mb 
 
start @nocallback 
plot @title = 'Gas Uptake GSA Setup', @xlabel = 'Time (hrs)', @ylabel = 'CP (ppm)', cp 
 
! Global sensitivity analysis, Morris method, response is cp 
gsa @method=morris @nlevels=10 @nsamps=50 @response=cp @coeffile='cp_gsa.csv' QPC QCC QLC QFC QSC BW VLC VFC VSC...  
PL PF PS PR PB KFC KLOS 
line @xvar=t @file='cp_gsa.csv' QPC QCC QLC QFC QSC BW VLC VFC VSC... 
PL PF PS PR PB KFC KLOS 


